


Application of Joints and Springs in ANSYS

Introduction

This tutorial was created using ANSYS 5.7.1. This tutorial will introduce:

- the use of multiple elements in ANSYS
- elements COMBIN7 (Joints) and COMBIN14 (Springs)
- obtaining/storing scalar information and store them as parameters.

A 1000N vertical load will be applied to a catapult as shown in the figure below. The catapult is built from steel tubing with an outer diameter of 40 mm, a wall thickness of 10, and a modulus of elasticity of 200GPa. The springs have a stiffness of 5 N/mm.



Preprocessing: Defining the Problem

1. Open preprocessor menu

/PREP7

2. Give example a Title

Utility Menu > File > Change Title ...

/title,Catapult

3. Define Element Types

For this problem, 3 types of elements are used: PIPE16, COMBIN7 (Revolute Joint), COMBIN14 (Spring-Damper) . It is therefore required that the types of elements are defined prior to creating the elements. This element has 6 degrees of freedom (translation along the X, Y and Z axis, and rotation about the X,Y and Z axis).

a. Define PIPE16

With 6 degrees of freedom, the PIPE16 element can be used to create the 3D structure.

- Preprocessor > Element Type > Add/Edit/Delete... > click 'Add'
- Select 'Pipe', 'Elast straight 16'
- Click on 'Apply' You should see 'Type 1 PIPE16' in the 'Element Types' window.

b. Define COMBIN7

COMBIN7 (Revolute Joint) will allow the catapult to rotate about nodes 1 and 2.

- Select 'Combination', 'Revolute Joint 7'
- Click 'Apply'.

c. Define COMBIN14

Now we will define the spring elements.

- Select 'Combination', 'Spring damper 14'
- Click on 'OK'

In the 'Element Types' window, there should now be three types of elements defined.

4. Define Real Constants

Real Constants must be defined for each of the 3 element types.

a. PIPE16

- Preprocessor > Real Constants > Add/Edit/Delete... > click 'Add'
- Select Type 1 PIPE16 and click 'OK'
- Enter the following properties, then click 'OK'

OD =
TKWALL = 10

40

b. 'Set 1' will now appear in the dialog box

c. COMBIN7 (Joint)

Five of the degrees of freedom (UX, UY, UZ, ROTX, and ROTY) can be constrained with different levels of flexibility. These can be defined by the 3 real constants: K1 (UX, UY), K2 (UZ) and K3 (ROTX, ROTY). For this example, we will use high values for K1 through K3 since we only expect the model to rotate about the Z axis.

- Click 'Add'
- Select 'Type 2 COMBIN7'. Click 'OK'.
- In the 'Real Constants for COMBIN7' window, enter the following geometric properties (then click 'OK'):

X-Y transnational stiffness K1: 1e9
Z directional stiffness K2: 1e9
Rotational stiffness K3: 1e9

- 'Set 2' will now appear in the dialog box.

Note: The constants that we define in this problem refer to the relationship between the coincident nodes. By having high values for the stiffness in the X-Y plane and along the Z axis, we are essentially constraining the two coincident nodes to each other.

d. COMBIN14 (Spring)

- Click 'Add'
- Select 'Type 3 COMBIN14'. Click 'OK'.
- Enter the following geometric properties:
Spring constant K: 5

e. In the 'Element Types' window, there should now be three types of elements defined.

5. Define Element Material Properties

1. Preprocessor > Material Props > Material Models
2. In the 'Define Material Model Behavior' Window, ensure that Material Model Number 1 is selected
3. Select Structural > Linear > Elastic > Isotropic
4. In the window that appears, enter the give the properties of Steel then click 'OK'.
Young's modulus EX: 200000
Poisson's Ratio PRXY: 0.33

6. Define Nodes

Preprocessor > (-Modeling-) Create > Nodes > In Active CS...
N, #, , x, y, z

We are going to define 13 Nodes for this structure as given in the following table (as depicted by the circled numbers in the figure above):

Node	Coordinates (x,y,z)
1	(0,0,0)
2	(0,0,1000)
3	(1000,0,1000)
4	(1000,0,0)
5	(0,1000,1000)
6	(0,1000,0)
7	(700,700,500)


8	(400,400,500)
9	(0,0,0)
10	(0,0,1000)
11	(0,0,500)
12	(0,0,1500)
13	(0,0,-500)

7. Create PIPE16 elements

Define element type

Preprocessor > (-Modeling-) Create > Elements > Elem Attributes ...

The following window will appear. Ensure that the 'Element type number' is set to 1 PIPE16, 'Material number' is set to 1, and 'Real constant set number' is set to 1. Then click 'OK'.



a. Create elements

Preprocessor > (-Modeling-) Create > Elements > (-Auto Numbered-) Thru Nodes

E, node a, node b


Create the following elements joining Nodes 'a' and Nodes 'b'.

Note: because it is difficult to graphically select the nodes you may wish to use the command line (for example, the first entry would be: **E,1,6**).

No de a	No de b
1	6
2	5
1	4
2	3
3	4
10	8
9	8
7	8
12	5
13	6

12	13
5	3
6	4

You should obtain the following geometry (Oblique view)



8. Create COMBIN7 (Joint) elements

Define element type

Preprocessor > (-Modeling-) Create > Elements > Elem Attributes
 Ensure that the 'Element type number' is set to 2 COMBIN7 and that 'Real constant set number' is set to 2. Then click 'OK'

a. Create elements

When defining a joint, three nodes are required. Two nodes are coincident at the point of rotation. The elements that connect to the joint must reference each of the coincident points. The other node for the joint defines the axis of rotation. The axis would be the line from the coincident nodes to the other node.

Preprocessor > (-Modeling-) Create > Elements > (-Auto Numbered-) Thru Nodes
 E,node a, node b, node c

Create the following lines joining Node 'a' and Node 'b'

Node a	Node b	Node c
1	9	11
2	10	11

9. Create COMBIN14 (Spring) elements

Define element type

Preprocessor > (-Modeling-) Create > Elements > Elem Attributes
 Ensure that the 'Element type number' is set to 3 COMBIN7 and that 'Real constant set number' is set to 3. Then click 'OK'

a. Create elements

Preprocessor > (-Modeling-) Create > Elements > (-Auto Numbered-) Thru Nodes
 E,node a, node b

Create the following lines joining Node 'a' and Node 'b'

Node a	Node b
5	8
8	6

10. **NOTE:** To ensure that the correct nodes were used to make the correct element in the above table, you can list all the elements defined in the model. To do this, select **Utilities Menu > List > Elements > Nodes + Attributes**.

11. Meshing


Because we have defined our model using nodes and elements, we do not need to mesh our model. If we initially defined our model using keypoints and lines, we would have had to create elements in our model by meshing the lines. It is the elements that ANSYS uses to solve the model.

12. Plot Elements

Utility Menu > Plot > Elements

You may also wish to turn on element numbering and turn off keypoint numbering

Utility Menu > PlotCtrls > Numbering ...



Solution Phase: Assigning Loads and Solving

1. Define Analysis Type

Solution > New Analysis > Static
ANTYPE, 0

2. Allow Large Deflection

Solution > Sol'n Controls > basic
NLGEOM, ON

Because the model is expected to deform considerably, we need to include the effects of large deformation.

3. Apply Constraints


Solution > (-Loads-) Apply > (-Structural-) > Displacement > On Nodes
o Fix Nodes 3, 4, 12, and 13. (ie - all degrees of freedom are constrained).

4. Apply Loads

Solution > (-Loads-) Apply > (-Structural-) > Force/Moment > On Nodes
o Apply a vertical point load of 1000N at node #7.

The applied loads and constraints should now appear as shown in the figure below.

Note: To have the constraints and loads appear each time you select 'Replot' in ANSYS, you must change some settings under **Utility Menu > Plot Ctrls > Symbols...** In the window that appears check the box beside 'All Applied BC's' in the 'Boundary Condition Symbol' section.



5. Solve the System


Solution > (-Solve-) Current LS
SOLVE

Note: During the solution, you will see a yellow warning window which states that the "Coefficient ratio exceeds 1.0e8". This warning indicates that the solution has relatively large displacements. This is due to the rotation about the joints.

Postprocessing: Viewing the Results

1. Plot Deformed Shape

General Postproc > Plot Results > Deformed Shape
PLDISP,2




2. Extracting Information as Parameters


In this problem, we would like to find the vertical displacement of node #7. We will do this using the GET command.

- a. Select **Utility Menu > Parameters > Get Scalar Data...**


- b. The following window will appear. Select 'Results data' and 'Nodal results' as shown then click 'OK'



- c. Fill in the 'Get Nodal Results Data' window as shown below:



- d. To view the defined parameter select **Utility Menu > Parameters > Scalar Parameters...**



Therefore the vertical displacement of Node 7 is 323.78 mm. This can be repeated for any of the other nodes you are interested in.

Command File Mode of Solution

The above example was solved using a mixture of the Graphical User Interface (or GUI) and the command language interface of ANSYS. This problem has also been solved using the ANSYS command language interface that you may want to browse. Copy and paste following code into Notepad or a similar text editor and save it to your computer. Now go to '**File > Read input from...**' and select the file.

```

/title, Catapult
/PREP7

ET,1,PIPE16          ! Element type 1
ET,2,COMBIN7         ! Element type 2
ET,3,COMBIN14        ! Element type 3

R,1,40,10           ! Real constants 1
R,2,1e9,1e9,1e9     ! Real constants 2
R,3,5, , ,          ! Real constants 3

MP,EX,1,200000      ! Young's modulus (Material 1)
MP,PRXY,1,0.33      ! Poisson's ratio (Material 1)

N, 1, 0, 0, 0       ! Node locations
N, 2, 0, 0,1000
N, 3,1000, 0,1000
N, 4,1000, 0, 0
N, 5, 0,1000,1000
N, 6, 0,1000, 0
N, 7, 700, 700, 500
N, 8, 400, 400, 500
N, 9, 0, 0, 0
N,10, 0, 0,1000
N,11, 0, 0, 500
N,12, 0, 0,1500
N,13, 0, 0,-500

TYPE,1              ! Turn on Element 1
REAL,1              ! Turn on Real constants 1
MAT,1               ! Turn on Material 1

E, 1, 6             ! Element connectivity
E, 2, 5
E, 1, 4
E, 2, 3
E, 3, 4
E,10, 8
E, 9, 8
E, 7, 8
E,12, 5
E,13, 6
E,12,13
E, 5, 3
E, 6, 4

TYPE,2              ! Turn on Element 2
REAL,2              ! Turn on Real constants 2

E, 1, 9, 11         ! Element connectivity
E, 2, 10, 11

```

^

```
TYPE,3           ! Turn on Element 3
REAL,3           ! Turn on Real constants 3

E,5,8            ! Element connectivity
E,8,6

/PNUM,KP,0       ! Number nodes
/PNUM,ELEM,1     ! Number elements
/REPLOT

FINISH
/SOLU            ! Enter solution phase

ANTYPE,0         ! Static analysis
NLGEOM,ON        ! Non-linear geometry on
NSUBST,5         ! 5 Load steps of equal size

D,3,ALL,0,,,4,12,13 ! Constrain nodes 3,4,12,13
F,7,FY,-1000     ! Load node 7

SOLVE
FINISH

/POST1

PLDISP,2
*GET,VERT7,NODE,7,U,Y
```

Reference

<http://www.mece.ualberta.ca/tutorials/ansys>